


Ejercicio 7

	LD F0, Cte5	
	ADDI R1, R0, #50	; longitud vectorial
	MTC1 VLR, R1	; VLR \leftarrow 50
Loop:	LV V1, Ra	; cargar fila del vector a
	LV V2, Rb	; cargar fila del vector b
	MULVV V1, V2, V1	; a[i, XXX] * b[i, XXX]
	ADDVS V1, V1, F0	; (a[i, XXX] * b[i, XXX]) + 5
	SV Rc, V1	; almacenar fila de c
	ADDI Ra, Ra, #50*8	
	ADDI Rb, Rb, #50*8	
	ADDI Rc, Rc, #50*8	
	SUBI R1, R1, #1	
	BNEZ R1, Loop	; si no ha finalizado el programa (50 iteraciones), volver a Loop

```
for (i=1;i<50;i++) {  
    for (j=0; j<50; j++){  
        c[i,j] = (a[i,j]*b[i,j])+5;  
    }  
}
```

b) Diagrama de tiempo de una iteración del bucle interno



Adaptar la fórmula para el cálculo de T_n :

Cuántas iteraciones se hacen? 50. Luego, en lugar de $\lceil n/MVL \rceil$ usamos la cte 50.

Además:

$n = n^\circ$ de elementos a procesar = $50 \times 50 = 2500$

$T_{\text{start}} = 1 + 12 + 7 + 6 + 12 = 38$

$T_{\text{chime}} = 2$

$$T_n = 50 \times (T_{\text{loop}} + T_{\text{start}}) + n \times T_{\text{chime}} = 50 \times (15 + 38) + 2500 \times 2 = 50 \times 53 + 5000 = 7650 \text{ ciclos}$$

c) Cuando el n° de filas crece indefinidamente ($n \rightarrow \infty$), el n° de iteraciones a realizar es siempre $n/50$. Luego:

$$T = (n/50) \times (15 + 38) + 2 \times n = 3,06 \times n$$

$$R = 2 \times n / 3,06 \times n = 0,654 \text{ FLOP/ciclo}$$


Ejercicio 10

a)	ADDI R1, R0, #50	; longitud vectorial
	ADDI R2, R0, #28	; nº de iteraciones
	LD F10, Cte5	; load 5 en F10
	ADDI Rc, Rc, #50*8	; Preparar puntero inicial a Rc
	MTC1 VLR, R1	; $VLR \leftarrow 50$
	ADDI Rs, R0, #29*8	; Stride para accesos a B (29*8 bytes)
Loop:	LVWS V1, (Rb, Rs)	; $V1 \leftarrow$ Columna de B
	LV V2, Ra	; $V2 \leftarrow$ fila de A
	MULVS V1, V1, F10	; $B*5$
	ADDVV V2, V2, V1	; $A+B*5$
	SV Rc, V2	; Store C
	ADDI Ra, Ra, #50*8	; Incrementar punteros
	ADDI Rb, Rb, #8	
	ADDI Rc, Rc, #50*8	
	SUBI R2, R2, #1	
	BNEZ R2, Loop	; Si no se ha acabado (28 iteraciones), volver a Loop

```
for (i=1; i < 29; i++) {  
    for (j=0; j<50; j++)  
         $c[i,j] = a[i-1,j] + b[j,i-1] * 5;$   
}
```

b) 16 bancos de memoria, tiempo de ciclo 2 ns

El stride en accesos a B es de 29 palabras: nº primo → 29 y 16 son primos entre sí. Luego, acceso sin conflictos ya que nº bancos de memoria (16) > tiempo acceso a memoria (12 ciclos)



Expresión de T_n : (ver ejercicio 7) Se hacen 28 iteraciones del bucle vectorial

$$T_n = (\text{nº iteraciones}) \times (T_{\text{loop}} + T_{\text{start}}) + n \times T_{\text{chime}} = 28 \times (15 + 12 + 7 + 6 + 12) + (28 \times 50) \times 2 = 28 \times 52 + 2800 = 4256 \text{ ciclos}$$

$$R = 2 \times 50 \times 28 \text{ FLOP} / 4256 \text{ ciclos} = 0,658 \text{ FLOP/ciclo}$$

$$t_c = 2 \text{ ns} = 2 \times 10^{-9} \text{ s} \rightarrow R = (0.658 \text{ FLOP/ciclo}) / (2 \times 10^{-9} \text{ s/ciclo}) = 329 \text{ MFLOPS}$$

c) Si el nº de filas de C y A, y el de columnas de B, crecen indefinidamente, el nº de iteraciones siempre será $n/50$

$$T_n = (\text{nº iteraciones}) \times (T_{\text{loop}} + T_{\text{start}}) + n \times T_{\text{chime}} = (n/50) \times (15 + 12 + 7 + 6 + 12) + 2n = (52n / 50) + 2n = 3,04n \text{ ciclos}$$

$$R = 2n / 3,04n = 0,658 \text{ FLOP/ciclo}$$

Se obtiene el mismo rendimiento, dado que la longitud total de los vectores siempre es múltiplo de 50